

VW-EQUATRAN
Modular Modeling Tool

例題集

2005/3/11

 **Omega Simulation Co.,Ltd.**
株式会社オメガシミュレーション

もくじ


1. アンモニア製造プロセスの物質収支計算	3
ステディステート（定常状態）シミュレーション	
物質収支計算	
ストリーム接続（総モル流量、モル組成を伝達）	
2. バッチ反応器のシミュレーション	9
ダイナミックシミュレーション	
信号線接続	
シーケンス記述（自動操作）	
3. シクロヘキサン製造プロセスの物質・熱収支計算	15
ステディステート（定常状態）シミュレーション	
物質収支・熱収支計算	
物性値ライブラリと物性関数の利用	
ストリーム接続（総モル流量、モル組成、温度、圧力を伝達）	

1. アンモニア製造プロセスの物質収支計算

プロセスの概念設計を想定した例として、アンモニア製造のプロセスの定常計算を考える。登場する装置は反応器と分離器のみというもっとも簡単なものである。なお、総モル流量、モル組成を伝達するストリームを使った例となる。

■問題

アンモニアは以下のように窒素と水素から直接合成される。



原料は乾燥した水素と窒素ガスで、不純物としてメタンを含む。反応器では 20MPa、500°C の条件で触媒の存在下で反応が行われる。反応器出のガスは冷却、凝縮により分離し、未反応ガスはリサイクルされて、液体アンモニアは製品として取り出される。また、未反応ガス中の不純物（メタン）の蓄積をパージにより防ぐ。

原料の組成（モル分率）がおのおの $\text{N}_2 : 0.24$ 、 $\text{H}_2 : 0.74$ 、 $\text{CH}_4 : 0.02$ で、その流量が 100 [kmol/h] の場合の物質収支を計算する。なお、リサイクル未反応ガスの 2% がパージされるものとする。




図 1.1 アンモニア製造プロセス

■考え方

プロセスの物質収支を求めるが、ここではその処理を大まかなブロックとして考える。反応器、分離器のほかに、合流点、分流点がある。また、バッテリーリミットとして、原料（フィード）と製品、パージがある。それぞれのブロックで入出の物質の収支式を記述し

て、全体としての物質収支を計算する。

ブロック間には値を伝達するものとしてストリームを定義する。ここでは物質収支のみを考えるので、総モル流量 (F) とモル組成 (x) を考えればよい。成分は N₂、H₂、NH₃、CH₄ の4成分を登録する。

問題から図 1-2 のようなブロック・フロー・ダイアグラムを作成する。s107 ではリサイクル指定が必要となる点に注意されたい。

VM アンモニア製造プロセス-1




図 1.2 ブロックフロー図

いま、総モル流量とモル組成のみを考えているので、バッテリーリミット (Feed、Prod、Purge) や合流点 (Mixer)、分流点 (Splitter) の記述はほぼ決まってくる。合流点や分流点では、総モル流量と成分モル流量の収支式を記述することになる。これについては次ページの囲みを参照されたい。

ユニット	変数	入出力	ソーステキスト本体
Feed	F : フィード量 [kmol/h]	INPUT	p1.F = F p1.x = x
	x : フィード組成 [-]	INPUT	
Prod / Purge	F : 流量 [kmol/h]	OUTPUT	F = f1.F x = f1.x
	x : 組成 [-]	OUTPUT	
Mixer			f1.F + f2.F = p1.F f1.F*f1.x + f2.F*f2.x = p1.F*p1.x
Splitter	r : パージ率 [-]	INPUT	f1.F = p1.F + p2.F f1.F*f1.x = p1.F*p1.x + p2.F*p2.x p1.F = f1.F * r p1.F*p1.x = f1.F*f1.x * r

物質収支の基本式

ブロックあるいはユニット単位で物質収支をとらえると、一般に単位時間あたり次式のようになる。

$$(\text{蓄積量}) = (\text{流入量}) - (\text{流出量}) + (\text{生成} \cdot \text{消滅量})$$

定常状態を扱う場合には時間あたりの変化を考えないので、左辺は 0 となり（蓄積量）の項は消去できる。

$$(\text{流入量}) - (\text{流出量}) + (\text{生成} \cdot \text{消滅量}) = 0$$


つぎに流入の数、流出の数、反応のある／なしで考えると、反応器、合流点、分流点／分離器の3タイプが基本となる。この際に総モル流量、成分流量を考慮するが、総モル流量は F_i 、モデル組成は x_{ij} で表すと、成分流量は $F_i x_{ij}$ となる。

■ 反応器（流入 1、流出 1）

生成・消滅を反応で代表しているが、他の場合（たとえば漏れ）でも同様に考えることができる。反応生成量の総モル量を R 、成分ごとの反応生成量を r_j で表す。もし、生成・消滅がない場合には $R = 0$ と考えればよい。

$$F_1 + R = F_2$$

$$F_1 x_{1,j} + r_j = F_2 x_{2,j}$$




■ 合流点（流入 2、流出 1）

合流点は単に 2 つの流入が加算されて 1 つの流出になる。

$$F_1 + F_2 = F_3$$

$$F_1 x_{1,j} + F_2 x_{2,j} = F_3 x_{3,j}$$



■ 分流点／分離器（流入 1、流出 2）


分流点および分離器では分配比 α を導入して分配割合を表現する。分離器の場合には成分ごとの分離比を指定するので、 α は 1 次元配列となる。

$$F_1 = F_2 + F_3$$

$$F_1 x_{1,j} = F_2 x_{2,j} + F_3 x_{3,j}$$

$$F_2 = F_1 \cdot \alpha$$

$$F_2 x_{2,j} = F_1 x_{1,j} \cdot \alpha$$



残る主要な装置である反応器と分離器をどのようにモデル化するかで、いろいろなアプローチが考えられよう。

■解法 1

はじめにもっとも簡単なモデルで解法してみる。

- ・ 反応器での反応量はフィードされた窒素の転化率で与える。
- ・ 分離器での製品への分離は、入ってくる成分流量に対する分離比で与える。

<反応器 (Reactor) >

転化率を `conv` として、種別: **INPUT**、型: **REAL** として定義する。ソーステキスト本体は反応量[kmol/h]を `r` とし、化学量論比を考慮して以下のようなになる。

```
VAR r(4) "反応量 [kmol/h]"

f1.F + SUM(r) = p1.F
f1.F*f1.x + r = p1.F*p1.x
-r(1) = f1.F*f1.x(1) * conv
-r(2) = -r(1) * 3
r(3) = -r(1) * 2
r(4) = 0
```


<分離器 (Separatr) >

分離比を `ratio` として、種別: **INPUT**、型: **COMP** として定義する。すなわち、`ratio` は成分ごとに考えるので配列変数となる。

```
f1.F = p1.F + p2.F
f1.F*f1.x = p1.F*p1.x + p2.F*p2.x
p1.F*p1.x = f1.F*f1.x * ratio
eq: SUM(p1.x) = 1
RESET p1.F#100[0, 10000] BY eq
```

■解法 1 の計算結果

転化率 `conv=0.16`、分離比 `ratio= (0.0006、0.00071、0.512、0.0010)` として定常計算を行う。リサイクルをデフォルトのままにしておくが、収束は緩やかで 450 回ほど繰り返すが必要となる。



■ 解法 2

解法 1 に比べてモデル精度を上げ、つぎのように反応モデル、分離モデルを考える。

- ・ 平衡反応を考慮し、化学平衡定数を与えて計算する。
- ・ 分離器では、アンモニアは分離比で与えるが、その他の気体がアンモニアにとけ込む溶解度[kmol/kmol]を与えて計算する。

< 反応器 (Reactor) >

化学平衡定数 K を、圧力を P として以下のように定義する。

$$K = K' P^2 = \frac{x_{NH_3}^2}{x_{N_2} x_{H_2}^3}$$

化学平衡定数を K として、種別: INPUT、型: REAL として定義する。また、化学量論から N 原子および H 原子について、入り出の原子数が等しいことを考慮して以下のソーステキストを作成する。

```
f1. F*f1. x(1)*2 + f1. F*f1. x(3) = p1. F*p1. x(1)*2 + p1. F*p1. x(3)
f1. F*f1. x(2)*2 + f1. F*f1. x(3)*3 = p1. F*p1. x(2)*2 + p1. F*p1. x(3)*3
f1. F*f1. x(4) = p1. F*p1. x(4)
eq2: p1. x(3)^2 = K*p1. x(1)*p1. x(2)^3
eq1: SUM(p1. x) = 1

RESET p1. x(1)#0.1[0,1] BY eq1
RESET p1. F#100[0,10000] BY eq2
```

< 分離器 (Separatr) >

分離比を $ratio$ (種別: INPUT、型: REAL) とし、溶解度を sol (種別: INPUT、型: COMP) と定義する。溶解度は、それぞれ N_2 : 0.00160、 H_2 : 0.00740、 CH_4 : 0.00225 というデー

タを用いる。

```


f1. F = p1. F + p2. F
f1. F*f1. x = p1. F*p1. x + p2. F*p2. x
p1. F*p1. x(3) = f1. F*f1. x(3) * ratio
p1. F*p1. x(1) = p1. F*p1. x(3) * sol(1)
p1. F*p1. x(2) = p1. F*p1. x(3) * sol(2)
p1. F*p1. x(4) = p1. F*p1. x(3) * sol(4)
eq: SUM(p1. x) = 1
RESET p1. F#1 BY eq

```

■解法 2 の計算結果

化学平衡定数 $K=0.35$ 、分離比 $ratio=0.512$ として定常計算を行う。この場合にも収束は緩やかで、360 回あまりの繰り返しが必要となる。

VM アンモニア製造プロセス-2



■実行の操作手順

1. エクスプローラを使って、Ammonia フォルダにある Ammonia1.vmp をダブルクリックして VM-EQUATRAN を起動する。
2. [ファイル]-[実行] () で実行モードに移る。
3. [設定]-[ループオーバー回数]で回数を 500 に変更する。
4. [実行]-[起動] () で起動し、定常計算ボタン をクリックして収束計算を始める。
5. [ファイル]-[編集] () で編集モードに戻り、[ファイル]-[開く] () で Ammonia2.vmp を読み込んで同様の手順で実行する。

2. バッチ反応器のシミュレーション

ダイナミックシミュレーションの例として、ジャケット付きバッチ反応器の運転を取り上げる。なお、この例題は EQUATRAN-G ユーザーズガイド [詳細編] の例題 15 に対応している。

ユニットとしては、反応器、ジャケット、PID 制御器を考え、データの受け渡しには信号線を使う。また、バッチプロセスには運転シーケンスが付き物であるが、自動操作の機能を使って実現する。

■問題

図 2.1 のようなバッチ反応器を用いて、原料 A および B から反応



によって製品 C を製造する。




図 2.1 バッチ反応器

つぎのシーケンスによって操作を行った場合のシミュレーションを考える。

- ①初期状態 …… 溶媒 $V=10\text{m}^3$ と原料 A 10kmol が装填されており反応器およびジャケットの温度は 20°C になっている。
- ②昇温 …… 熱媒 (120°C) を用いて反応器の温度 T が 90°C になるまで昇温する。
- ③反応 …… 原料 B を図 2-2 に示すようなパターンで、積算量 30kmol になるまで供給する。反応器の温度は 90°C に制御するが、最初の 0.2h の間は手動でジャケット温度を 88°C に設定するものとする。
- ④保温 …… 原料の供給および、冷媒の供給を停止したまま、 0.5h 保温して反応を進行させる。
- ⑤冷却 …… 冷媒 (20°C) で反応器の温度が 40°C になるまで冷却する。




図 2.2 運転シーケンス

計算条件は次のとおりとする。

- ・ 反応速度 r [kmol/h m³]

$$r = k \cdot C_A \cdot C_B^3$$

$$k = A \cdot \exp\left(\frac{-E}{RT}\right)$$

ただし、 $A=2.43 \times 10^{15}$ 、 $E=83700$ 、 $R=8.3145$ であり、 C_A 、 C_B は A および B の濃度 [kmol/m³]である。

- ・ 反応熱 $\Delta H = -50200$ kJ/kmol
- ・ 溶媒の比熱 $c_p = 1.86$ kJ/kg°C
- ・ 溶媒の密度 $\rho = 900$ kg/m³
- ・ ジャケットの伝熱面積 $A_j = 20$ m²
- ・ ジャケットの伝熱係数 $U_j = 580$ W/m² K
- ・ PID 制御器 TC1 のパラメータ (制御動作は逆動作)
 - 比例帯 $PB = 50$ %
 - 積分時間 $T_I = 1000$ s
 - 微分時間 $T_D = 360$ s
- ・ ジャケット温度 T_j °C

なお、ジャケット温度は TC2 の設定値 T_{jset} に対して、時定数 360 s の一次おくれで表わせるものとする。

■考え方

各ユニットには以下のような外部変数を指定して接続先と信号線で結ぶ。

	変数名	内容	接続先
反応器 (Resctor)	FB	B のフィード量	Bfeed. FB
	Tj	ジャケット温度	Jacket. Tj
ジャケット (Jacket)	Tjset	ジャケット温度設定値	PID. OUT
PID 制御器 (PID)	IN	入力値	Reactor. T

原料 B のフィード量は 10kmol/h² のスピードでフィード量を増やし、20kmol/h で一定量にしてから、累計 30kmol/h で止める。このためにこのフィード量を計算するフィード量の調整というユニットを用意した。

VM バッチ反応器




図 2.3 ブロックフロー図

■各ユニットモデル

<フィード量の調整>

10kmol/h² のスピードでフィード量を増やし、20kmol/h で一定量にしてから、累計 30kmol/h で止めるという操作を表現するためのソーステキストを作成した。FBon はフィード用スイッチで、1 になるとフィードが開始される。

```

FB = min(t1*10, 20) * (FBon==1)
t1' = 1; t1#0
CHANGE t1#0 ON (FBon==1)
FBtotal' = FB
    
```

<反応器>

問題に与えられた反応速度式を使い、濃度変化を微分方程式で表現した。ここで、FB は B

のフィード量である。

$$\frac{C_A}{dt} = -r$$

$$\frac{C_B}{dt} = -3 \cdot r + \frac{F_B}{V}$$

$$\frac{C_c}{dt} = r$$

反応器の温度はつぎの熱収支式で求められる。

$$V\rho c_p \frac{dT}{dt} = Q_r - Q_j$$

ここで Q_r は反応熱、 Q_j はジャケットによる除熱量であり、つぎのように与えられる。

$$Q_r = \Delta H V r$$

$$Q_j = 3.6 \cdot U_j A_j (T - T_j)$$

```
/* 反応 */
CA' = -r
CB' = -3*r + FB/V
CC' = r
r = k*CA*CB^3
k = 2.43e15*exp(-83700/8.3145/(T+273.15))

/* 熱収支 */
V*rho*cp*T' = Qr - Qj
Qr = -dH*V*r
Qj = 3.6*Uj*Aj*(T-Tj)

OUTPUT _time, CA, CB, CC, T, Tj STEP 0.01
```

<ジャケット>

```
/* 1次遅れ要素 */
Tj' = (Tjset-Tj)/(tau/3600)
```

<PID 制御器>

MAN モード、AUTO モードを切り替えて動作でき、制御動作（正動作／逆動作）の指定が可能な汎用の PID 制御器ユニットに仕上げた。PID 制御器の出力は次式で計算される。

$$OUT = K_p \left(e + \frac{1}{T_I} \int e dt + T_D \frac{de}{dt} \right)$$

$$K_p = \frac{100}{PB}$$

$$e = PV - SV$$

```

PV = IN
TIh = TI/3600
TDh = TD/3600
e = PV-SV
U0 = MV*TIh*PB/100
    = -MV*TIh*PB/100 WHEN INCDEC == 1
U1 # 0
CHANGE U1 # U0 ON (MODE!=1)
U1' = e
    = 0 WHEN MODE == 1
U2 = 100/PB*(e+U1/TIh+TDh*DERIV(e, 0))
U3 = U2
    = -U2 WHEN INCDEC == 1
OUT = U3
    = MV WHEN MODE == 1

```

<自動操作>

動かしながら画面から値を変更することも可能だが、問題に与えられた運転シーケンスを実現するために、以下のような操作表を作成した。PID 制御器の制御モード **MODE** や操作値 **MV**、設定値 **SV** とフィード用 **SW** である **FBon** に値を書き込んでいる。

```

[time == 00:01:00]
    PID.MV = 120
[time >= +00:00:00]
& [Reactor.T >= 90]
    Bfeed.FBon = 1
    PID.MV = 88
[time >= +00:12:00]
    PID.MODE = 2
    PID.SV = 90
[time >= +00:00:00]
& [Bfeed.FBtotal >= 30]
    Bfeed.FBon = 0
    PID.MODE = 1
    PID.MV = 90
[time >= +00:30:00]
    PID.MODE = 1
    PID.MV = 20
[time >= +00:00:00]

```

& [Reactor.T <= 40]
FREEZE “反応終了”

■計算結果

64 倍速で約 3 時間半に相当するシミュレーションを行った。




図 2.4 計算結果

■実行の操作手順

1. エクスプローラを使って、BatchReact フォルダにある BatchReact.vmp をダブルクリックして VM-EQUATRAN を起動する。
2. [ファイル]-[実行] () で実行モードに移る。
3. [ファイル]-[読み込み] () で初期状態ファイル AutoOper.vmi を開く。
4. [モニター]-[トレンド]-[表示] () で2つのトレンドを表示するようにしておく。
5. タイムスケールを「実時間」から「64 倍速」に変更する。
6. 運転ボタン で自動実行する。


3. シクロヘキサン製造プロセスの物質・熱収支計算

「アンモニア製造プロセス」の例題に比べて、熱収支やフラッシュ計算を取り入れた、より精密な定常状態シミュレーションである。ストリームによるデータの受け渡しには、総モル流量 F 、モル組成 x のほかに、温度 T 、圧力 P も考慮している（ストリーム定義でこの4つの変数にチェックを入れておく）。また、物性計算では各成分に対する物性値を標準物性ライブラリから読み込んで利用し、エンタルピ計算やフラッシュ計算に物性計算関数を利用した。

（注）標準物性ライブラリや物性計算関数を使うためには、物性計算オプションが必要となる。

■問題

シクロヘキサン (C_6H_{12}) は以下のようにベンゼン (C_6H_6) の水素添加反応で合成される。ここでは、主要装置を考慮して、メインのフローを考えることにする。



液体ベンゼンと補給水素は、リサイクル水素、リサイクル液と混合して反応器にフィードされる。補給水素には不純物として窒素が含まれる。反応器では、2.1MPa、200°Cの条件下で触媒を用いて水素添加反応を行う。反応器出の飽和蒸気は冷却したのち、フラッシュドラムで気液に分離する。気相は一部をパージ (8.2%) し、未反応水素はコンプレッサで加圧してリサイクルする。液相はさらに減圧してフラッシュ分離して製品シクロヘキサンを取り出す。また液相の一部はリサイクルする。




図 3.1 シクロヘキサン製造プロセス

原料ベンゼンは 41.83kmol/h、補給水素は 128.84kmol/h（窒素が 0.296mol%含まれる）で供給される場合の物質収支、熱収支を計算せよ。なお、反応器ではフィードされたベンゼンの 99.8%が反応する（転化率が 0.998）ものとする。

出典：W.D.Seider、J.D.Seider、D.R.Lewin 著『Process Design Principles』P.232（1999）

■ 考え方

物質収支計算の考え方は「アンモニア製造プロセス」の例題と同様である。

ここでさらに考慮したのは温度と圧力である。各成分の物性値に物性ライブラリを、温度（エンタルピー）計算やフラッシュ計算には物性計算関数を利用した。物性計算関数の引数については『VM-EQUATRAN チュートリアル』Appendix Bに一覧があるので参照されたい。

まず、成分として H_2 、 N_2 、ベンゼン、シクロヘキサンをライブラリから選択し、計算法は水素を含む高圧系なので **SRK** 式を採用した。

問題から図 3.2 のようなプロセスフロー図（PFD）を作成した。今回は各ユニットについてそれぞれユニット図形を作成して登録した。




図 3.2 シクロヘキサン製造プロセスのプロセスフロー図

< 反応器 RA01 >

反応器では転化率と反応条件（温度、圧力）が与えられるので以下のようなになる。


```

VAR xfl(_NC), xfl0(_NC), xfv(_NC), xfv0(_NC)
    , xpl(_NC), xpl0(_NC), xpv(_NC), xpv0(_NC), Dh(_NC)

/* 物質収支 */
f1.F + SUM(r) = p1.F
f1.F*f1.x + r = p1.F*p1.x
-r(1) = -r(3) * 3
r(2) = 0
-r(3) = f1.F*f1.x(3) * conv /* conv:INPUT */
r(4) = -r(3)

/* 出口温度・圧力 */
T = p1.T /* T:INPUT */
P = p1.P /* P:INPUT */

/* 除熱量 */
PHYHCLV( 1, f1.T, f1.P, 1, f1.x, hfl, rfl, xfl, hfv, rfv, xfv, hf )
/* 入口エンタルピ */
Hf = hf*1000
PHYHCLV( 1, p1.T, p1.P, 1, p1.x, hpl, rpl, xpl, hpv, rpv, xpv, hp )
/* 出口エンタルピ */
Hp = hp*1000
PHYGTDHV( 1, Dh ) /* 標準生成熱 */
Qr = (Dh(4)-Dh(3)-Dh(1)*3)*1000
Q = Hp*p1.F - Hf*f1.F - Qr*r(3)

```

反応での生成・消滅量の計算は「アンモニア製造プロセス」の例題と同様なので省略する。反応熱で発生する熱量は、フィード側とプロダクト側のエンタルピと、標準生成熱から計算する。そのために、物性計算関数 `PHYHCLV()` を使って、フィード側とプロダクト側の気液混相のエンタルピを求め、`PHYGTDHV()` で成分ごとの標準生成熱を得る。

気液混相のエンタルピ計算を行う `PHYHCLV()` では、温度 `f1.T`、圧力 `f1.P`、組成 `f1.x` を与えると、液相エンタルピ `hfl`、液相率 `rfl`、液相組成 `xfl`、気相エンタルピ `hfv`、気相率 `rfv`、気相組成 `xfv`、混相エンタルピ `hf` が得られる。ただし、使用するのは混相エンタルピのみである。

標準生成熱は `PHYGTDHV()` を呼ぶことで、成分ごとの値が1次元配列で引数に渡される。これらを使って反応器で除熱する熱量を計算している。

<クーラー CW01>

出口の温度、圧力が与えられている。除熱量を計算するために、入口側のエンタルピを温度から換算する関数 `PHYCLHVP()` で、出口側は気液混相のエンタルピ計算をする `PHYHCLV()` を使う。

```

/* 物質収支 */
f1.x = p1.x
f1.F = p1.F

/* 出口温度・圧力 */
T = p1.T      /* T: INPUT */
P = p1.P      /* P: INPUT */

/* 除熱量 */
PHYCLHVP( 1, f1.T, f1.P, f1.x, hf ) /* 入口エンタルピ */
PHYHCLV( 1, p1.T, p1.P, 1, f1.x, hpl, rl, xl, hpv, rv, xv, hp ) /* 出口エンタルピ */

Hf = hf*1000
Hp = hp*1000
Q = ( Hp - Hf ) * p1.F

/* 圧損 */
dP = p1.P - f1.P

```

<フラッシュドラム FA01/FA02>

フラッシュ計算は物性計算関数 PHYFLSH() を使用すれば容易に求められる。このときに与える温度、圧力は、フィード側の温度、圧力を使う。

```

/* 物質収支 */
f1.F = p1.F + p2.F
p1.F/f1.F = rv
PHYFLSH( 1, f1.T, f1.P, f1.x, rv, p2.x, p1.x ) /* フラッシュ計算 */

/* 入口温度・圧力 */
f1.T = T
f1.P = P

/* 出口温度・圧力 */
T = p1.T
T = p2.T
P = p1.P
P = p2.P

```

<減圧弁 VT01>

出口圧力のみが与えられており、出口温度を入口条件のエンタルピから PHYTCLV() を使って計算する。

```

VAR xpl(_NC) , xpv(_NC)

/* 出口状態量 */

```

```

f1.F = p1.F
f1.x = p1.x
T = p1.T
P = p1.P /* Pは INPUT */
PHYCLHLQ( 1, f1.T, f1.P, f1.x, hf ) /* 入口エンタルピ */
PHYTCLV( 1, P, hf, 1, f1.x, hpl, rpl, xpl, hpv, rpv, xpv, T )
/* エンタルピ→温度 */
f1.P = p1.P + dP

```

<ポンプ GA01/GA02>

加える動力によるエンタルピ変化を考慮して、PHYTCL()で出口の温度をエンタルピから換算する。

```

/* 物質収支 */
f1.F = p1.F
f1.x = p1.x

/* 熱収支 */
PHYCLHLQ( 1, f1.T, f1.P, f1.x, hf ) /* 温度→エンタルピ */
hp = hf + 3.6 * Power / f1.F /* Power: INPUT */
PHYTCL( 1, p1.P, hp, p1.x, T ) /* エンタルピ→温度 */

/* 出口温度・圧力 */
T = p1.T
P = p1.P /* P: INPUT */
f1.P + dP = p1.P

```

<コンプレッサ GB01>

ポンプと同様なので省略する。ただし、動力からの変換では機械効率を加味してある。

<合流点 MX01>

入口側の条件から PHYCLHVP()、PHYCLHLQ()でエンタルピを計算し、PHYTCLV()で出口側の温度に変換している。

```

VAR x(_NC)

/* 物質収支 */
f1.F + f2.F + f3.F + f4.F = p1.F
f1.F*f1.x + f2.F*f2.x + f3.F*f3.x + f4.F*f4.x = p1.F*x
x/SUM(x) = p1.x

/* 熱収支 */
PHYCLHVP( 1, f1.T, f1.P, f1.x, h1 ) /* 入口1エンタルピ */
PHYCLHVP( 1, f2.T, f2.P, f2.x, h2 ) /* 入口2エンタルピ */

```

```

PHYCLHLQ( 1, f3.T, f3.P, f3.x, h3 ) /* 入口3エンタルピ */
PHYCLHLQ( 1, f4.T, f4.P, f4.x, h4 ) /* 入口4エンタルピ */
ht = ( h1*f1.F+h2*f2.F+h3*f3.F+h4*f4.F )/ p1.F
PHYTCLV( 1, P, ht, 1, p1.x, hl, rl, xl, hv, rv, xv, T )
/* エンタルピ→温度 */

/* 入口圧力 */
f1.P = P

/* 出口温度・圧力 */
P = p1.P
T = p1.T

```

<分流点 SP01/SP02>

分流比の考え方は「アンモニア製造プロセス」の例題と同様。フィード側の温度、圧力でプロダクト側に伝える。

```


/* 物質収支 */
f1.F = p1.F + p2.F
f1.F*f1.x = p1.F*p1.x + p2.F*p2.x
p1.F = f1.F * R /* R: INPUT */
p1.F*p1.x = f1.F*f1.x * R

/* 出口温度・圧力 */
f1.T = p1.T
f1.T = p2.T
f1.P = p1.P
f1.P = p2.P




```

■ 計算結果

シクロヘキサン製造プロセス



■ 実行の操作手順

1. エクスプローラを使って、Cycrohexane フォルダにある Cycrohex.vmp をダブルクリックして VM-EQUATRAN を起動する。
2. [ファイル]-[実行] () で実行モードに移る。
3. [設定]-[ループオーバー回数]で回数を 200 に変更する。
4. [実行]-[起動] () で起動し、定常計算ボタン () をクリックして収束計算を始める。